

Some thoughts about note–duration representation in the ABC language

Part 2

1. Introduction

In 'Some thoughts about note–duration representation in the ABC language' (1) I have presented the MCM language as a slight modification of the ABC language. The following can be said:

1. The MCM–language is more close to the standard musical notation than the ABC–language. It means that visualizing the MCM–code to a mental image of music notation is much easier. Thus the code is more readable and comprehensive (2). In this text I will try to convince you with two examples.
2. However, the MCM language is less compact in comparison to ABC language. Compactness could be enhanced by applying Lilypond's approach to make rhythm 'sticky' (3). Example: C2CCC is the same as C2C2C2C2
3. The MCM–language is context–independent (C16 is always a sixteenth note C etc.), which means that entry of MCM–code is less error–prone.

Does the MCM–language has more advantages than disadvantages, compared to ABC? To answer this question in an objective way, let me present you two unison pieces from Josquin Des Prez and Olivier Messiaen (4). Two remarks on these music fragments :

1. Josquin's and Messiaen's music have no bars: this is historically correct.
2. Messiaen's music has an additive rhythm. ABC and MCM (with no automatic beaming) have the advantage to notate this easily!

First, I show you some ABC and MCM code, then the score. Read the code and ask yourself the question:

Can I make a match between code and notation in an easy and fast way?

And if you like this question also:

Can I translate the code to inner music in an easy and fast way?

Then, check your mental image and evaluate the code in terms of readability and comprehensiveness (5).

2. Example 1: Messiaen, Corps glorieux (1939)

X:1
T:ABC with L: 1/16
M:none
L:1/16
K:C clef=treble
(D4 C2 ^F2^GB2 A^G^F=E D2=F2A_E2 ^C=GA^G D8)

X:2
T:ABC with L: 1/8
M:none
L:1/8
K:C clef=treble
(D2 C ^F^G/B A/^G/^F/=E/ D=FA/_E ^C/=G/A/^G/ D8)

X:3
T:MCM (current implementation)
M:none
L:mcm_default
K:C clef=treble
(D4 C8 ^F8^G16B8 A16^G16^F16=E16 D8=F8A16_E8 ^C16=G16A16^G16 D2)

X:4
T:Olivier Messiaen, Corps Glorieux (1939)
T:MCM (future Lilypond like ?)
M:none
L:mcm_default
K:C clef=treble
(D4 C8 ^F8^G16B8 A16^G^F=E D8=F8A16_E8 ^C16=GA^G D2)

Olivier Messiaen, Corps Glorieux (1939)



3. Example 2: Josquin Des Prez, Pange lingua (ca. 1515)

X:1
T:ABC with L: 1/4
M:none
L:1/4
K:C
c4 c4 d4 c6 A2 B4 A4 c3 B G2 B2 A2 c3 B G A B c d2 c2 A4 G4 "^#"F2 G8 |

X:2
T:ABC with L: 1/2
M:none
L:1/2
K:C
c2 c2 d2 c3 A B2 A2 c3/2 B/ G B A c3/2 B/ G/ A/ B/ c/ d c A2 G2 "^#"F G4 |

X:3
T:ABC with L: 1/1
M:none
L:1/1
K:C
c c d c3/2 A/ B A c3/4 B/4 G/ B/ A/ c3/4 B/4 G/4 A/4 B/4 c/4 d/ c/ A G "^#"F/ G2 |

X:4
T:MCM (current implementation)
M:none
L:mcm_default
K:C
c1 c1 d1 c1. A2 B1 A1 c2. B4 G2 B2 A2 c2. B4 G4 A4 B4 c4 d2 c2 A1 G1 "^#"F2 G11 |

X:5
T:MCM (Future Lilypond like ?)
M:none
L:mcm_default
K:C
c1 c d c1. A2 B1 A c2. B4 G2 B A c2. B4 G A B c d2 c A1 G "^#"F2 G11 |



4. Concluding remarks

Does the MCM–language has more advantages than disadvantages, compared to ABC? I think that the answer on this question depends on how you use ABC or MCM. My adaptation of ABC has to do with my personal goals and my goals as an educator:

1. to read MCM code as fast as a score of Beethoven etc.
2. to imagine the sound from MCM code as fast as from a score of Beethoven etc.
3. to correct MCM code as fast as a score in e.g. Finale or Sibelius

Having said that, I conclude with the following remarks:

In my opinion ABC works well for rhythmic undifferentiated, unison melodies in a divide rhythm such as many of the 67.000 tunes on www.abcmusicnotation.com. However, a MCM adaptation will have comparable compact code.

Example (ABC with L:1/2) :

A | F G A c | _B2 A

equals (future MCM ?)

A2 | F G A c | _B1 A2

The present MCM – implementation leads indeed to less compact code

A2 | F2 G2 A2 c2 | _B1 A2

However, it is comprehensive and thus it works for me...

Footnotes

- (1) www.mcmusiceditor.com/download/maliepaard_ABC-note-duration-representation.pdf , v.1.1 – 16 August 2010
- (2) The remark of an abcuser "All music notation is a relative one. There is no such thing as an absolute basic notation value." is in this context completely irrelevant. I do not speak about musical performance but about musical notation!
- (3) In Lilypond C16 C C C C C C C represents eight sixteenth notes, in two groups (beams automated). I think that for better readability this should be better: C16CCC C16CCC. So I think about implementing this feature in the MC Musiceditor, only for groups of 4 (3?) or more eighth, sixteenth and thirty–second notes (because my last MCM implementation of Josquin Des Prez' example doesn't convince me).
- (4) These examples can easily be extended with two–part and more part pieces. Imagine what this means in terms of readability.
- (5) The next weeks I will evaluate these codes in the same way with my international music students.

About the author:

Reinier Maliepaard is psychologist, software engineer, organist and teacher at the ArtEZ Conservatorium Netherlands (music theory and music history).

His freeware music notation programm MC Musiceditor (Windows) can be downloaded at www.mcmusiceditor.com

These notes have been typeset with MC Musiceditor 6.0.4